@echo off
setlocal enabledelayedexpansion

REM 定义要遍历的目标目录为src
set "TARGET_DIR=src"

REM 检查src目录是否存在
if not exist "%TARGET_DIR%\" (
    echo 错误：src目录不存在！
    exit /b 1
)

REM 获取当前目录的上一级目录名称作为输出文件名
for %%I in ("%cd%") do set "parent_dir=%%~nxI"

REM 设置输出文件名为上一级目录名称加上 .pri 扩展名
set OUTPUT_FILE=%parent_dir%.pro

REM 清空或创建目标文件
> %OUTPUT_FILE% echo.

REM 写入固定的头部信息
echo TEMPLATE = subdirs >> %OUTPUT_FILE%

REM 遍历src目录下的所有子文件夹（不递归）
for /d %%D in ("%TARGET_DIR%\*") do (
  set "folder=%%~nxD"
  echo SUBDIRS += $$PWD/%TARGET_DIR%/!folder! >> %OUTPUT_FILE%
)

REM 添加最后一部分固定内容
echo CONFIG += ordered >> %OUTPUT_FILE%

REM 提示用户文件已创建
echo .pri file has been created: %OUTPUT_FILE%

endlocal
